RESTXQ

This page presents one of the Web Application services. It describes how to use the RESTXQ API of BaseX.

RESTXQ, introduced by Adam Retter, is an API that facilitates the use of XQuery as a server-side processing language for the Web. RESTXQ has been inspired by Java’s JAX-RS API: it defines a pre-defined set of XQuery 3.0 annotations for mapping HTTP requests to XQuery functions, which in turn generate and return HTTP responses.

Please note that BaseX provides various extensions to the original draft of the specification:


 * Multipart types are supported, including multipart/form-data
 * A %rest:error annotation can be used to catch XQuery errors
 * Servlet errors can be redirected to other RESTXQ pages
 * A RESTXQ Module provides some helper functions
 * Parameters are implicitly cast to the type of the function argument
 * The Path Annotation can contain regular expressions
 * annotations, support for input-specific content-type parameters
 * annotation to cancel running RESTXQ functions
 * Quality factors in the Accept header will be evaluated
 * Support for server-side quality factors in the  annotation
 * Better support for the OPTIONS and HEAD methods

=Introduction=

Preliminaries
The RESTXQ service is accessible via http://localhost:8984/.

All RESTXQ annotations are assigned to the  namespace, which is statically bound to the rest prefix. A Resource Function is an XQuery function that has been marked up with RESTXQ annotations. When an HTTP request comes in, a resource function will be invoked that matches the constraints indicated by its annotations.

If a RESTXQ URL is requested, the module directory and its sub-directories will be traversed, and all XQuery files will be parsed for functions with RESTXQ annotations. Sub-directories that include an .ignore file will be skipped. In addition, XQuery modules that cannot be parsed will be ignored if is enabled.

To speed up processing, the functions of the existing XQuery modules are automatically cached in main memory:
 * Functions will be invalidated and parsed again if the timestamp of their module changes.
 * File monitoring can be adjusted via the option. In productive environments with a high load, it may be recommendable to change the timeout, or completely disable monitoring.
 * If files are replaced while the web server is running, the RESTXQ module cache should be explicitly invalidated by calling the static root path /.init or by calling the rest:init function.

Examples
A first RESTXQ function is shown below:

 module namespace page = 'http://basex.org/examples/web-page';

declare %rest:path("hello/{$who}") %rest:GET function page:hello($who) { Hello { $who }! };

If the URI http://localhost:8984/hello/World is accessed, the result will be:

 &lt;response&gt; &lt;title&gt;Hello World!&lt;/title&gt; &lt;/response&gt;

The next function demonstrates a POST request:

 declare %rest:path("/form") %rest:POST %rest:form-param("message","{$message}", "(no message)") %rest:header-param("User-Agent", "{$agent}") function page:hello-postman( $message as xs:string,  $agent   as xs:string* ) as element(response) { &lt;response type='form'&gt; &lt;message&gt;{ $message }&lt;/message&gt; &lt;user-agent&gt;{ $agent }&lt;/user-agent&gt; &lt;/response&gt; };

If you post something (e.g. using curl or the embedded form at http://localhost:8984/)...

 curl -i -X POST --data "message='CONTENT'" http://localhost:8984/form

...you will receive something similar to the following result:

 HTTP/1.1 200 OK Content-Type: application/xml; charset=UTF-8 Content-Length: 107 Server: Jetty(8.1.11.v20130520)

 'CONTENT' curl/7.31.0

=Request=

This section shows how annotations are used to handle and process HTTP requests.

Constraints
Constraints restrict the HTTP requests that a resource function may process.

Paths
A resource function must have a single Path Annotation with a single string as argument. The function will be called if a URL matches the path segments and templates of the argument. Path templates contain variables in curly brackets, and map the corresponding segments of the request path to the arguments of the resource function. The first slash in the path is optional.

The following example contains a path annotation with three segments and two templates. One of the function arguments is further specified with a data type, which means that the value for  will be cast to an   before being bound:

 declare %rest:path("/a/path/{$with}/some/{$variable}") function page:test($with, $variable as xs:integer) { ... };

Variables can be enhanced by regular expressions:

 (: Matches all paths with "app" as first, a number as second, and "order" as third segment :) declare %rest:path("app/{$code=[0-9]+}/order") function page:order($code) { ... };

(: Matches all other all paths starting with "app/" :) declare %rest:path("app/{$path=.+}") function page:others($path) { ... };

If multiple path candidates are found for the request, the one with more segments will be preferred.

Content Negotiation
Two following annotations can be used to restrict functions to specific content types:

%rest:consumes("application/xml", "text/xml") %rest:produces("application/atom+xml")
 * HTTP Content Types: A function will only be invoked if the HTTP Content-Type header of the request matches one of the given content types. Example:
 * HTTP Accept: A function will only be invoked if the HTTP Accept header of the request matches one of the defined content types. Example:

By default, both content types are */*. Quality factors supplied by a client will also be considered in the path selection process. If a client supplies the following accept header…


 * /*;q=0.5,text/html;q=1.0

…and if two RESTXQ functions exist with the same path annotation, one with the produces annotation, and another with  , the second function will be called, because the quality factor for   documents is highest.

Server-side quality factors are supported as well: If multiple function candidates are left over after the above steps, the  parameter will be considered. The function with the highest quality factor will be favored:

 %rest:produces("text/html;qs=1") %rest:produces("*/*;qs=0.8")

Note that the annotation will not affect the content type of the actual response. You will need to supply an additional  annotation.

Default Methods
The HTTP method annotations are equivalent to all HTTP request methods except TRACE and CONNECT. Zero or more methods may be used on a function; if none is specified, the function will be invoked for each method.

The following function will be called if GET or POST is used as request method:

 declare %rest:GET %rest:POST %rest:path("/post") function page:post { "This was a GET or POST request" };

The POST and PUT annotations may optionally take a string literal in order to map the HTTP request body to a function argument. Once again, the target variable must be embraced by curly brackets:

 declare %rest:PUT("{$body}") %rest:path("/put") function page:put($body) { "Request body: " || $body };

Custom Methods
Custom HTTP methods can be specified with the %rest:method annotation. An optional body variable can be supplied as second argument:

 declare %rest:path("binary-size") %rest:method("SIZE", "{$body}") function page:patch( $body  as xs:base64Binary ) { "Request method: " || request:method, "Size of body: " || bin:length($body) };

If an OPTIONS request is received, and if no function is defined, an automatic response will be generated, which includes an  header with all supported methods.

If a HEAD request is received, and if no function is defined, the corresponding GET function will be processed, but the response body will be discarded.

Content Types
The body of a POST or PUT request will be converted to an XQuery item. Conversion can be controlled by specifying a content type. It can be further influenced by specifying additional content-type parameters:

For example, if  is specified as content type, the input will be transformed to JSON, and the lax QName conversion rules will be applied, as described in the JSON Module.

Input options
Conversion options for, and  can also be specified via annotations with the   prefix. The following function interprets the input as text with the CP1252 encoding and treats the first line as header:

<pre class="brush:xquery"> declare %rest:path("/store.csv") %rest:POST("{$csv}") %input:csv("header=true,encoding=CP1252") function page:store-csv($csv as document-node) { "Number of rows: " || count($csv/csv/record) };

Multipart Types
The single parts of a multipart message are represented as a sequence, and each part is converted to an XQuery item as described in the last paragraph.

A function that is capable of handling multipart types is identical to other RESTXQ functions:

<pre class="brush:xquery"> declare %rest:path("/multipart") %rest:POST("{$data}") %rest:consumes("multipart/mixed") (: optional :) function page:multipart($data as item*) { "Number of items: " || count($data) };

Parameters
The following annotations can be used to bind request values to function arguments. Values will implicitly be cast to the type of the argument.

Query Parameters
The value of the first parameter, if found in the query component, will be assigned to the variable specified as second parameter. If no value is specified in the HTTP request, all additional parameters will be bound to the variable (if no additional parameter is given, an empty sequence will be bound):

<pre class="brush:xquery"> declare %rest:path("/params") %rest:query-param("id", "{$id}") %rest:query-param("add", "{$add}", 42, 43, 44) function page:params($id as xs:string?, $add as xs:integer+) { <result id="{ $id }" sum="{ sum($add) }"/> };

HTML Form Fields
Form parameters are specified the same way as query parameters. Their values are the result of HTML forms submitted with the content type.

<pre class="brush:xquery"> %rest:form-param("parameter", "{$value}", "default")

File Uploads
Files can be uploaded to the server by using the content type multipart/form-data (the HTML5 multiple attribute enables the upload of multiple files):

<pre class="brush:xml"> <form action="/upload" method="POST" enctype="multipart/form-data"> <input type="file" name="files" multiple="multiple"/>

The file contents are placed in a map, with the filename serving as key. The following example shows how uploaded files can be stored in a temporary directory:

<pre class="brush:xquery"> declare %rest:POST %rest:path("/upload") %rest:form-param("files", "{$files}") function page:upload($files) { for $name   in map:keys($files) let $content := $files($name) let $path   := file:temp-dir || $name return (   file:write-binary($path, $content),    <file name="{ $name }" size="{ file:size($path) }"/>  ) };

HTTP Headers
Header parameters are specified the same way as query parameters:

<pre class="brush:xquery"> %rest:header-param("User-Agent", "{$user-agent}") %rest:header-param("Referer", "{$referer}", "none")

Cookies
Cookie parameters are specified the same way as query parameters:

<pre class="brush:xquery"> %rest:cookie-param("username", "{$user}") %rest:cookie-param("authentication", "{$auth}", "no_auth")

Query Execution
In many RESTXQ search scenarios, input from browser forms is processed and search results are returned. User experience can generally be made more interactive if an updated search request is triggered with each key click. However, this may lead to many expensive parallel requests, from which only the result of the last request will be relevant for the client.

With the  annotation, it can be enforced that only one instance of a function will be executed for the same client. If the same function will be called for the second time, the already running query will be stopped, and the HTTP error code 460 will be returned instead:

<pre class="brush:xquery"> (: If fast enough, returns the result. Otherwise, if called again, raises 460 :) declare %rest:path("/search") %rest:query-param("term", "{$term}") %rest:single function page:search($term as xs:string) { <ul>{ for $result in db:open('large-db')//*[text = $term] return <li>{ $result }</li> }</ul> };

By specifying a string along with the annotation, functions can be bundled together, and one request can be canceled by calling another one.

This is shown by another example, in which the first function can be interrupted by the second one. If you call both functions in separate browser tabs, you will note that the first tab will return, and the second one will return stopped.

<pre class="brush:xquery"> declare %rest:path("/compute") %rest:single("EXPENSIVE") function local:compute { (1 to 100000000000000)[. = 0] };

declare %rest:path("/stop") %rest:single("EXPENSIVE") function local:stop { stopped };

The following things should be noted:


 * If a query will be canceled, there will be no undesirable side-effects. For example, it won’t be possible to kill a query if it is currenly updating the database or perfoming any other I/O operations. As a result, the termination of a running query can take some more time as expected.
 * The currently executed function is bound to the current session. This way, a client will not be able to cancel requests from other clients. As a result, functions can only be stopped if there was at least one previous successful response, in which initial session data was returned to the client.

=Response=

By default, a successful request is answered with the HTTP status code 200 (OK) and is followed by the given content. An erroneous request leads to an error code and an optional error message (e.g. 404 for “resource not found”).

Custom Response
Custom responses can be generated in XQuery by returning an  element, an   child node that matches the syntax of the EXPath HTTP Client Module specification, and optional child nodes that will be serialized as usual. A function that yields a response on an unknown resource may look as follows:

<pre class="brush:xquery"> declare %output:method("text") %rest:path("") function page:error404 { <rest:response> <http:response status="404"> <http:header name="Content-Language" value="en"/> <http:header name="Content-Type" value="text/plain; charset=utf-8"/> </http:response> </rest:response>, "The requested resource is not available." };

Redirects
rest:redirect element.

The server can invite the client (e.g., the web browser) to make a second request to another URL by sending a 302 response:

<pre class="brush:xml"> <rest:response> <http:response status="302"> <http:header name="Location" value="new-location"/> </http:response> </rest:response>

The convenience function $$ can be called to create such a response.

In the XQuery context, redirects are particularly helpful if Updates are performed. An updating request may send a redirect to a second function that generates a success message, or evaluates an updated database:

<pre class="brush:xquery"> declare %updating %rest:path('/app/init') function local:create { db:create('app',, 'root.xml'), db:output(web:redirect('/app/ok')) };

declare %rest:path('/app/ok') function local:ok { 'Stored documents: ' || count(db:open('app')) };

Forwards
A server-side redirect is called forwarding. It reduces traffic among client and server, and the forwarding will not change the URL seen from the client’s perspective:

<pre class="brush:xml"> <rest:forward>new-location</rest:forward>

The fragment can also be created with the convenience function $$.

Output
The content-type of a response can be influenced by the user via Serialization Parameters. The steps are described in the REST chapter. In RESTXQ, serialization parameters can be specified in the query prolog, via annotations, or within the REST response element:

Query Prolog
In main modules, serialization parameters may be specified in the query prolog. These parameters will then apply to all functions in a module. In the following example, the content type of the response is overwritten with the media-type parameter:

<pre class="brush:xquery"> declare option output:media-type 'text/plain';

declare %rest:path("version1") function page:version1 { 'Keep it simple, stupid' };

Annotations
Global serialization parameters can be overwritten via  annotations. The following example serializes XML nodes as JSON, using the JsonML format:

<pre class="brush:xquery"> declare %rest:path("cities") %output:method("json") %output:json("format=jsonml") function page:cities { element cities { db:open('factbook')//city/name } };

The next function, when called, generates XHTML headers, and text/html will be set as content type:

<pre class="brush:xquery"> declare %rest:path("done") %output:method("xhtml") %output:omit-xml-declaration("no") %output:doctype-public("-//W3C//DTD XHTML 1.0 Transitional//EN") %output:doctype-system("http://www.w3.org/TR/xhtml1/DTD/xhtml1-transitional.dtd") function page:html { <html xmlns="http://www.w3.org/1999/xhtml"> done };

Response Element
Serialization parameters can also be specified in a REST reponse element in a query. Serialization parameters will be overwritten:

<pre class="brush:xquery"> declare %rest:path("version3") function page:version3 { <rest:response> <output:serialization-parameters> <output:media-type value='text/plain'/> </output:serialization-parameters> </rest:response>, 'Not that simple anymore' };

=Error Handling=

Raise Errors


If an error is raised during the evaluation of a RESTXQ function, an HTTP response with the status code 400 is generated. The response body contains the full error message and stack trace.

With the $$ function, you can abort query evaluation and enforce a premature HTTP response with the supplied status code and response body text:

<pre class="brush:xquery"> declare %rest:path("/teapot") function page:teapot { web:error(418, "I'm a pretty teapot") };

The XQuery error code and the stack trace will be suppressed in the body of the HTTP response.

Catch XQuery Errors
XQuery runtime errors can be processed via error annotations. Error annotations have one or more arguments, which represent the error codes to be caught. The codes equal the names of the XQuery 3.0 try/catch construct:

All error codes that are specified for a function must have the same precedence. The following rules apply when catching errors:


 * Codes with a higher precedence (smaller number) will be given preference.
 * A global RESTXQ error will be raised if two functions with conflicting codes are found.

Similar to try/catch, the pre-defined variables (code, description, value, module, line-number, column-number, additional) can be bound to variables via error parameter annotations, which are specified the same way as query parameters.

Errors may occur unexpectedly. However, they can also be triggered by a query, as demonstrated by the following example:

<pre class="brush:xquery"> declare %rest:path("/check/{$user}") function page:check($user) { if($user = ('jack', 'lisa')) then 'User exists' else fn:error(xs:QName('err:user'), $user) };

declare %rest:error("err:user") %rest:error-param("description", "{$user}") function page:user-error($user) { 'User "' || $user || '" is unknown' };

Catch HTTP Errors
Errors that occur outside RESTXQ can be caught by adding error-page elements with an error code and a target location to the web.xml configuration file (find more details in the Jetty Documentation):

<pre class="brush:xml"> <error-page> <error-code>404</error-code> /error404 </error-page>

The target location may be another RESTXQ function. The request:attribute function can be used to request details on the caught error:

<pre class="brush:xquery"> declare %rest:path("/error404") function page:error404 { "URL: " || request:attribute("javax.servlet.error.request_uri") || ", " || "Error message: " || request:attribute("javax.servlet.error.message") };

=User Authentication=

If you want to provide restricted access to parts of a web applications, you will need to check permissions before returning a response to the client. The Permissions layer is a nice abstraction for defining permission checks.

=Functions=

The Request Module contains functions for accessing data related to the current HTTP request. Two modules exist for setting and retrieving server-side session data of the current user (Session Module) and all users known to the HTTP server (Sessions Module). The RESTXQ Module provides functions for requesting RESTXQ base URIs and generating a WADL description of all services. Please note that the namespaces of all of these modules must be explicitly specified via module imports in the query prolog.

The following example returns the current host name:

<pre class="brush:xquery"> import module namespace request = "http://exquery.org/ns/request";

declare %rest:path("/host-name") function page:host { 'Remote host name: ' || request:remote-hostname };

=References=

Documentation:


 * RESTXQ Specification, First Draft
 * RESTful XQuery, Standardised XQuery 3.0 Annotations for REST. Paper, XMLPrague, 2012
 * RESTXQ. Slides, MarkLogic User Group London, 2012
 * Web Application Development. Slides from XMLPrague 2013

Examples:


 * Sample code combining XQuery and JavaScript: Materials and paper from Amanda Galtman, Balisage 2016.
 * DBA: The Database Administration interface, bundled with the full distributions of BaseX.

=Changelog=


 * Version 9.3


 * Updated: Custom Methods: Better support for the OPTIONS and HEAD methods.
 * Updated: XQuery Errors: Suppress stack trace and error code in the HTTP response.
 * Removed: rest:redirect element ($$ can be used instead)


 * Version 9.2


 * Updated: Ignore XQuery modules that cannot be parsed


 * Version 9.0


 * Added: Support for server-side quality factors in the  annotation
 * Updated: Status code 410 was replaced with 460
 * Removed: restxq prefix


 * Version 8.4


 * Added:  annotation


 * Version 8.1


 * Added: support for input-specific content-type parameters
 * Added:  annotations


 * Version 8.0


 * Added: Support for regular expresssions in the Path Annotation
 * Added: Evaluation of quality factors that are supplied in the Accept header


 * Version 7.9


 * Updated: XQuery Errors, extended error annotations
 * Added: %rest:method


 * Version 7.7


 * Added: Error Handling, File Uploads, Multipart Types
 * Updated: RESTXQ function may now also be specified in main modules (suffix: *.xq).
 * Updated: the RESTXQ prefix has been changed from restxq to rest.
 * Updated: parameters are implicitly cast to the type of the function argument
 * Updated: the RESTXQ root url has been changed to http://localhost:8984/


 * Version 7.5


 * Added: new XML elements <rest:redirect/> and <rest:forward/>